Mat 1. 2-timersprøve den 14. maj 2018.

JE 9.5.18

Opgave 1

```
> restart:with(LinearAlgebra):with(plots):
```

En reel funktion f af to reelle variable er givet ved

```
> f:=(x,y)->4*y*(x^2+1/3*y^2-1);
```

$$f := (x, y) \mapsto 4\, y \left(x^2 + \frac{y^2}{3} - 1\right)$$

```
> expand(f(x,y));
```

$$4\, y\, x^2 + \frac{4}{3}\, y^3 - 4\, y$$

Spørgsmål 1

Hvis f har lokalt ekstremum i et punkt, så må punktet være et stationært punkt, da f ikke har undtagelsespunkter.

$\nabla f(x, y) = \left(f'_x(x, y), f'_y(x, y) \right) = (8\,xy, 4\,x^2 + 4\,y^2 - 4) = (0, 0) \Leftrightarrow$

$8xy = 0$ og $4x^2 + 4\,y^2 - 4 = 0$ $\Leftrightarrow x = 0$ og $4y^2 - 4 = 0$ eller $y = 0$ og $4x^2 - 4 = 0$ $\Leftrightarrow x = 0$ og $y = \pm 1$ eller $y = 0$ og $x = \pm 1$.

Samtlige stationære punkter for f er da $(1, 0)$, $(-1, 0)$, $(0, 1)$ og $(0, -1)$.

Hessematricen for f i punktet (x, y) er

```
> H(x,y):=<diff(f(x,y),x,x),diff(f(x,y),y,x)|diff(f(x,y),x,y),
  diff(f(x,y),y,y)>;
```

$$H(x, y) := \begin{bmatrix} 8\, y & 8\, x \\ 8\, x & 8\, y \end{bmatrix}$$

```
> H(1,0):=subs(x=1,y=0,H(x,y));
```

$$H(1, 0) := \begin{bmatrix} 0 & 8 \\ 8 & 0 \end{bmatrix}$$

```
> Eigenvalues(H(1,0),output=list);
```

$$[8, -8]$$

Da de to egenværdier for $\mathbf{H}(1,0)$ har modsat fortegn, så har f ikke lokalt ekstremum i det stationære punkt (1,0) (saddelpunkt).

```
> H(-1,0):=subs(x=-1,y=0,H(x,y));
```

$$H(-1, 0) := \begin{bmatrix} 0 & -8 \\ -8 & 0 \end{bmatrix}$$

```
> Eigenvalues(H(-1,0),output=list);
```

$$[8, -8]$$

Da de to egenværdier for $\mathbf{H}(-1,0)$ har modsat fortegn, så har f ikke lokalt ekstremum i det stationære punkt (−1,0) (saddelpunkt).

```
> H(0,1):=subs(x=0,y=1,H(x,y));
```

$$H(0,1) := \begin{bmatrix} 8 & 0 \\ 0 & 8 \end{bmatrix}$$

```
> Eigenvalues(H(0,1),output=list);
```

$$[8, 8]$$

Da begge egenværdierne for **H**(0,1) er positive, har f egentligt lokalt minimum i det stationære punkt (0, 1) med værdien

```
> 'f(0,1)'=f(0,1);
```

$$f(0,1) = -\frac{8}{3}$$

```
> H(0,-1):=subs(x=0,y=-1,H(x,y));
```

$$H(0,-1) := \begin{bmatrix} -8 & 0 \\ 0 & -8 \end{bmatrix}$$

```
> Eigenvalues(H(0,-1),output=list);
```

$$[-8, -8]$$

Da begge egenværdierne for **H**(0, −1) er negative, har f egentligt lokalt maksimum i det stationære punkt (0, −1) med værdien

```
> 'f(0,-1)'=f(0,-1);
```

$$f(0,-1) = \frac{8}{3}$$

```
> contourplot(f(x,y),x=-3..3,y=-3..3,contours=80);
```



En ellipse E i (x, y)-planen er givet ved ligningen $x^2 + \frac{y^2}{3} - 1 = 0 \Leftrightarrow \frac{x^2}{1} + \frac{y^2}{3} = 1$.

Spørgsmål 2

Ellipsen E har centrum i $(0, 0)$, halve storakse $a = \sqrt{3}$ og halve lilleakse $b = 1$.

Spørgsmål 3

Vi betragter den begrænsede og afsluttede punktmængde M, der omkrandses af E.

```
> implicitplot(x^2+y^2/3=1,x=-2..2,y=-2..2,scaling=constrained,
  filled=true,coloring=[gray,white], title="M");
```



Da M er begrænset og afsluttet og da f er kontinuert i M, så har f et globalt minimum og et globalt maksimum i M. Da f ikke har undtagelsespunkter i det indre af M, så antages disse værdier enten i et stationært punkt i det indre af M eller på randen af M.
De eneste stationære punkter i det indre af M er $(0, 1)$ og $(0, -1)$ og

```
> 'f(0,1)'=f(0,1);'f(0,-1)'=f(0,-1);
```

$$f(0, 1) = -\frac{8}{3}$$

$$f(0, -1) = \frac{8}{3}$$

Da randen E af M er givet ved ligningen $x^2 + \frac{y^2}{3} - 1 = 0$, så er $f(x, y) = 0$ for alle $(x, y) \in E$.

Ved nummerisk sammenligning af disse undersøgelser fås, at det globale minimum er $-\frac{8}{3}$, som antages i punktet $(0, 1)$ og at det globale maksimum er $\frac{8}{3}$, som antages i punktet $(0, -1)$.

Det bemærkes, at da M er sammenhængende, så er værdimængden $f(M) = [-\frac{8}{3}; \frac{8}{3}]$.

```
> plot3d(f(x,y),x=-2..2,y=-2..2,view=-3..3);
```



Opgave 2

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
```

I (x, y)-planen betragtes punktmængden $B = \{ (x, y) \mid -1 \le x \le 1 \text{ og } 0 \le y \le \cosh(x) \}$.

```
> plot(cosh(x),x=-1..1,scaling=constrained,filled=true,title="M");
```



Spørgsmål 1

En parameterfremstilling for B er

$$\mathbf{r}(u,v) = \begin{bmatrix} u \\ v\cosh(u) \end{bmatrix}, u \in [-1;1], v \in [0;1].$$

En højdefunktion h defineret på (x, y)-planen er givet ved $z = h(x, y)$, hvor

```
> h:=(x,y)->2-x;
```

$$h := (x, y) \mapsto 2 - x$$

Lad F betegne den del af grafen for h som ligger lodret over B.

Spørgsmål 2

En parameterfremstilling for F er

```
> r:=<u,v*cosh(u),h(u,v*cosh(u))>;
```

$$r := \begin{bmatrix} u \\ v\cosh(u) \\ 2-u \end{bmatrix}$$

hvor $u \in [-1;1]$ og $v \in [0;1]$.

```
> ru:=diff~(r,u);
```

$$ru := \begin{bmatrix} 1 \\ v\sinh(u) \\ -1 \end{bmatrix}$$

```
> rv:=diff~(r,v);
```

$$rv := \begin{bmatrix} 0 \\ \cosh(u) \\ 0 \end{bmatrix}$$

Fladens normalvektor er

```
> N:=kryds(ru,rv);
```

$$N := \begin{bmatrix} \cosh(u) \\ 0 \\ \cosh(u) \end{bmatrix}$$

Den til **r** hørende Jacobi-funktion er

```
> Jacobi:=sqrt(prik(N,N)) assuming cosh(u)>0;
```

$$Jacobi := \sqrt{2}\,\cosh(u)$$

Spørgsmål 3

```
> f:=(x,y,z)->(z-1)/sqrt(2);
```

$$f := (x,y,z) \mapsto \frac{z-1}{\sqrt{2}}$$

$$\int_F \frac{(z-1)}{\sqrt{2}}\,\mathrm{d}\mu = \int_{v=0}^{1}\int_{u=-1}^{1} \frac{(z(u,v)-1)}{\sqrt{2}}\,\mathrm{Jacobi}(u,v)\,\mathrm{d}u\,\mathrm{d}v = \int_{v=0}^{1}\int_{u=-1}^{1} (1-u)\cosh(u)\,\mathrm{d}u\,\mathrm{d}v\,.$$

```
> integranden:=f(vop(r))*Jacobi;
```

$$integranden := (1-u)\cosh(u)$$

```
> Int(Int(integranden,u=-1..1),v=0..1)=int(int(integranden,u=
  -1..1),v=0..1);
```

$$\int_0^1\int_{-1}^{1} (1-u)\cosh(u)\,\mathrm{d}u\,\mathrm{d}v = 2\sinh(1)$$

Opgave 3

```
> restart:with(LinearAlgebra):with(plots):
  vop:=proc(X) op(convert(X,list)) end proc:
```

Det oplyses, at

$$\begin{bmatrix} x'(t) \\ y'(t) \end{bmatrix} = \begin{bmatrix} 4\,x(t) - 10\,y(t) \\ 2\,x(t) - 5\,y(t) \end{bmatrix}, t \in \mathbb{R},$$

har den fuldstændige løsning

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 \mathrm{e}^{-t} \\ 2\,c_1 + c_2 \mathrm{e}^{-t} \end{bmatrix}, t \in \mathbb{R}\,, c_1, c_2 \in \mathbb{R},$$

I (x, y)-planen betragtes vektorfeltet $\mathbf{V}(x, y) = \begin{bmatrix} 4\,x - 10\,y \\ 2\,x - 5\,y \end{bmatrix}$.

Spørgsmål 1

Den ovenfor oplyste fuldstændige løsning er netop flowkurverne for det i (x, y)-planen betragtede vektorfelt $\mathbf{V}(x, y) = \begin{bmatrix} 4\,x - 10\,y \\ 2\,x - 5\,y \end{bmatrix}$.

Til bestemmelse af konstanterne c_1 og c_2 så $\begin{bmatrix} x(0) \\ x(0) \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$ haves det lineære ligningssystem

$$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 \\ 2\,c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 5 & 2 \\ 2 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$

med tilhørende totalmatrix

```
> T:=<<5,2>|<2,1>|<1,1>>;
```

$$T := \begin{bmatrix} 5 & 2 & 1 \\ 2 & 1 & 1 \end{bmatrix}$$

```
> c:=LinearSolve(T);
```

$$c := \begin{bmatrix} -1 \\ 3 \end{bmatrix}$$

Den søgte flowkurve er da

$$\begin{bmatrix} x(t) \\ y(t) \end{bmatrix} = \begin{bmatrix} -5 + 6\mathrm{e}^{-t} \\ -2 + 3\mathrm{e}^{-t} \end{bmatrix}, \mathrm{t} \in \mathbb{R}.$$

L er er det rette liniestykke fra punktet (1, 1) til punktet (2, 2).

Spørgsmål 2

En parameterfremstilling for L er

$$\begin{bmatrix} x(u) \\ y(u) \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}, u \in [1;2].$$

Spørgsmål 3

Til bestemmelse af konstanterne c_1 og c_2 så $\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}$ haves det lineære ligningssystem

$$\begin{bmatrix} x(0) \\ y(0) \end{bmatrix} = \begin{bmatrix} 5\,c_1 + 2\,c_2 \\ 2\,c_1 + c_2 \end{bmatrix} = \begin{bmatrix} 5 & 2 \\ 2 & 1 \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} u \\ u \end{bmatrix}$$

med tilhørende totalmatrix

```
> T:=<<5,2>|<2,1>|<u,u>>;
```

$$T := \begin{bmatrix} 5 & 2 & u \\ 2 & 1 & u \end{bmatrix}$$

```
> c:=LinearSolve(T);
```

$$c := \begin{bmatrix} -u \\ 3\,u \end{bmatrix}$$

Den flowkurve $\mathbf{r}(u, t)$ for $\mathbf{V}$, der opfylder $\mathbf{r}(u, 0) = \begin{bmatrix} u \\ u \end{bmatrix}$ har da parameterfremstillingen

```
> r:=(u,t)-><-5*u+6*u*exp(-t),-2*u+3*u*exp(-t)>:
> r(u,t);
```

$$\begin{bmatrix} -5\,u + 6\,u\,e^{-t} \\ -2\,u + 3\,u\,e^{-t} \end{bmatrix}$$

For $t = 1$ og $u \in [1; 2]$ fås

```
> r(u,1);
```

$$\begin{bmatrix} -5\,u + 6\,u\,e^{-1} \\ -2\,u + 3\,u\,e^{-1} \end{bmatrix}$$

som er en parameterfremstilling for den kurve, som L er blevet deformeret til, til tiden t = 1.

Da $\mathbf{r}(u, 1)$ kan skrives som $\mathbf{r}(u, 1) = u \begin{bmatrix} -5 + 6e^{-1} \\ -2 + 3e^{-1} \end{bmatrix}$, hvor $u \in [1; 2]$, så er kurven det rette liniestykke fra punktet $(-5 + 6e^{-1}, -2 + 3e^{-1})$ til punktet $(-10 + 12e^{-1}, -4 + 6e^{-1})$.

```
> L:=plot(x,x=1..2,thickness=5,color=blue):
> P1:=plot([vop(r(u,1)),u=1..2],thickness=5,color=red):
> K1:=plot([vop(r(1,t)),t=0..1],linestyle=4):
> K2:=plot([vop(r(2,t)),t=0..1],linestyle=4):
> display(L,P1,K1,K2,scaling=constrained);
```



Opgave 4

```
> restart:with(plots):
  prik:=(x,y)->VectorCalculus[DotProduct](x,y):
  kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector):
  vop:=proc(X) op(convert(X,list)) end proc:
  grad:=X->convert(Student[VectorCalculus][Del](X),Vector):
  div:=V->VectorCalculus[Divergence](V):
  rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)  end
  proc:
```

I (x, y, z)-rummet er der givet vektorfeltet

```
> V:=(x,y,z)-><3*x-y^2,4*y*z,y-2*z^2>:
> V(x,y,z);
```

$$\begin{bmatrix} -y^2+3x \\ 4yz \\ -2z^2+y \end{bmatrix}$$

Ω er en massiv omdrejningscylinder med bundradius 1, x-aksen som symmetriakse og afgrænset af planerne $x = 0$ og $x = 1$ (se figuren i opgaveteksten).

Spørgsmål 1

$\partial\Omega$ er den lukkede overflade af Ω orienteret med udadrettet enhedsnormalvektor.
Af Gauss' sætning fås da

```
> div(V)(x,y,z);
```

$$3$$

$$\mathrm{Flux}(\mathbf{V}, \partial\Omega) = \int_\Omega \mathrm{Div}(\mathbf{V})\ d\mu = \int_\Omega 3\ d\mu = 3\mathrm{Vol}(\Omega) = 3\pi.$$

Betragt den cirkelskive C i planen $x = 1$, som afslutter cylinderen i x-aksens retning og dens randkurve ∂C (vist med rød på figuren i opgaveteksten).

Spørgsmål 2

Parameterfremstilling for C

```
> r:=<1,u*cos(v),u*sin(v)>;
```

$$r := \begin{bmatrix} 1 \\ u\cos(v) \\ u\sin(v) \end{bmatrix}$$

hvor u $\in$ [0;1] og v $\in$ [0;2π].

```
> ru:=diff~(r,u);
```

$$ru := \begin{bmatrix} 0 \\ \cos(v) \\ \sin(v) \end{bmatrix}$$

```
> rv:=diff~(r,v);
```

$$rv := \begin{bmatrix} 0 \\ -u\sin(v) \\ u\cos(v) \end{bmatrix}$$

Fladens normalvektor

```
> N:=simplify(kryds(ru,rv));
```

$$N := \begin{bmatrix} u \\ 0 \\ 0 \end{bmatrix}$$

hvor $u \in]0;1]$, danner netop højreskrue med den valgte orientering af den lukkede randkurve ∂C for C (vist med blå tangentvektor på figuren i opgaveteksten).
Den angivne parameterfremstilling for C opfylder derfor betingelsen.

Spørgsmål 3

Af Stokes' sætning fås da

$$\mathrm{Cirk}(\mathbf{V},\partial C) = \int_{\partial C} \mathbf{V} \cdot \mathbf{e}_{\partial C}\, \mathrm{d}\mu = \mathrm{Flux}(\mathbf{Rot}(\mathbf{V}),C) = \int_{C} \mathbf{n}_C \,.\, \mathbf{Rot}(\mathbf{V})\, \mathrm{d}\mu =$$

$$\int_{v=0}^{2\pi}\int_{u=0}^{1} \mathbf{N}(u,v) \,.\, \mathbf{Rot}(\mathbf{V})\,(\mathbf{r}(u,v))\, \mathrm{d}u\, \mathrm{d}v\,.$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 1-4\,y \\ 0 \\ 2\,y \end{bmatrix}$$

Rotationen taget på fladen

```
> Rot:=rot(V)(vop(r));
```

$$Rot := \begin{bmatrix} 1-4\,u\cos(v) \\ 0 \\ 2\,u\cos(v) \end{bmatrix}$$

```
> integranden:=prik(N,Rot);
```

$$integranden := u\,(1-4\,u\cos(v))$$

```
> Int(Int(integranden,u=0..1),v=0..2*Pi)=int(int(integranden,u=
  0..1),v=0..2*Pi);
```

$$\int_0^{2\pi}\int_0^{1} u\,(1-4\,u\cos(v))\,\mathrm{d}u\,\mathrm{d}v = \pi$$